Taylor series

Exercise 1.1

(a)

```
> y := ln(sqrt((1+x)/(1-x)));
```

$$y := \frac{1}{2} \ln\left(\frac{1+x}{1-x}\right) \quad (1)$$

```
> simplify(diff(y,x$5));
```

$$-\frac{24 \left(1 + 10 x^2 + 5 x^4\right)}{(1+x)^5 (x-1)^5} \quad (2)$$

```
> subs(x=0,%);
```

$$24 \quad (3)$$

```
> %/5!;
```

$$\frac{1}{5} \quad (4)$$

(b)

```
> simplify(subs(x=0,y));
```

$$0 \quad (5)$$

```
> subs(x=0,simplify(diff(y,x$1)))/1!;
```

$$1 \quad (6)$$

```
> subs(x=0,simplify(diff(y,x$2)))/2!;
```

$$0 \quad (7)$$

```
> subs(x=0,simplify(diff(y,x$3)))/3!;
```

$$\frac{1}{3} \quad (8)$$

```
> subs(x=0,simplify(diff(y,x$4)))/4!;
```

$$0 \quad (9)$$

```
> subs(x=0,simplify(diff(y,x$5)))/5!;
```

$$\frac{1}{5} \quad (10)$$

(c)

```
> a := (n) -> subs(x=0,diff(y,x$n))/n!;
```

$$a := n \rightarrow \frac{subs\left(x = 0, \frac{\partial^n}{\partial x^n} y\right)}{n!} \quad (11)$$

```
> seq(a(n),n=1..5);
```

$$1, 0, \frac{1}{3}, 0, \frac{1}{5} \quad (12)$$

(d)

```
> add(a(k)*x^k,k=1..12);
```

$$x+\frac{1}{3}x^3+\frac{1}{5}x^5+\frac{1}{7}x^7+\frac{1}{9}x^9+\frac{1}{11}x^{11} \quad (13)$$

```
> series(y,x=0,13);
```

$$x+\frac{1}{3}x^3+\frac{1}{5}x^5+\frac{1}{7}x^7+\frac{1}{9}x^9+\frac{1}{11}x^{11}+\mathrm{O}(x^{13}) \quad (14)$$

(e)

The obvious guess is that y is the sum of $\frac{x^n}{n}$ for all odd integers n, or in other words $y=\sum_{k=0}^{\infty}\frac{x^{2k+1}}{2k+1}$.

This can be checked as follows:

```
> sum(x^(2*k+1)/(2*k+1),k=0..infinity);
```

$$\frac{1}{2}\ln\left(\frac{1+x}{1-x}\right) \quad (15)$$

```
> simplify(%-y,symbolic);
```

$$0 \quad (16)$$

Exercise 1.2

(a)

```
> s := convert(series(sin(x),x=0,12),polynom);
```

$$s := x-\frac{1}{6}x^3+\frac{1}{120}x^5-\frac{1}{5040}x^7+\frac{1}{362880}x^9-\frac{1}{39916800}x^{11} \quad (17)$$

(b)

```
> plot([s,sin(x)],x=-8..8,-10..10);
```



(c)

```
> t := (n) -> convert(series(sin(x),x=0,n),polynom);
```

$$t := n \rightarrow convert(series(\sin(x), x=0, n), polynom) \quad (18)$$

```
> plot([t(14),sin(x)],x=-8..8,-10..10);
```



The graph of $t(n)$ (for x in $[-8, 8]$) is visually indistinguishable from that of $\sin(x)$ when $20 \leq n$.

```
> plot([t(20),sin(x)],x=-8..8,-10..10);
```



However, the two functions diverge sharply for x outside this range:

```
> plot([t(20),sin(x)],x=-10..10,-10..10);
```



(d)

```
> plot([sin(x),seq(t(4*n+2),n=0..10)],x=0..20,-2..10);
```




```
> plot([sin(x),seq(t(4*n+4),n=0..10)],x=0..20,-10..2);
```



We do not bother with $t(n)$ for odd n, because $t(n) = t(n-1)$ for odd n.

(e)

```
> r := (n) -> convert(series(cos(x),x=0,n),polynom);
```

$$r := n \rightarrow convert(series(\cos(x), x = 0, n), polynom) \quad \textbf{(19)}$$

```
> expand(t(10)^2+r(10)^2);
```

$$\frac{1}{60963840} x^{14} + \frac{1}{1814400} x^{10} - \frac{1}{4354560} x^{12} - \frac{1}{2090188800} x^{16} + \frac{1}{131681894400} x^{18} + 1 \quad \textbf{(20)}$$

Note that $t(10)$ is an approximation to $\sin(x)$ and $r(10)$ is an approximation to $\cos(x)$, so $t(10)^2 + r(10)^2$ should be an approximation to $\sin(x)^2 + \cos(x)^2 = 1$. In fact we see that $t(10)^2 + r(10)^2$ is 1 plus a sum of terms like $\frac{x^k}{N}$ where k is at least 10 and N is at least 1000000. If x is not too big, then all these extra terms will be very small, so $t(10)^2 + r(10)^2$ will indeed be close to 1. To see this graphically, we plot $1 - t(10)^2 - r(10)^2$:

```
> plot(1-t(10)^2-r(10)^2,x=-5..5,-1..1);
```



The error is very small for x between -3 and 3, but it grows very big when $x < -5$ or $5 < x$.

Exercise 1.3

```
> series(x*(1+x)/(1-x)^3,x=0,11);
```

$$x + 4x^2 + 9x^3 + 16x^4 + 25x^5 + 36x^6 + 49x^7 + 64x^8 + 81x^9 + 100x^{10} + O(x^{11}) \quad \textbf{(21)}$$

Note that the coefficient of x^3 is $9 = 3^2$, the coefficient of x^4 is $16 = 4^2$ and so on. It should be clear from this that the series is $\sum_{k=1}^{\infty} k^2 x^k$. We can ask Maple to confirm this as follows:

```
> sum(k^2*x^k,k=1..infinity);
```

$$-\frac{x(1+x)}{(x-1)^3} \quad \textbf{(22)}$$

This is the same as $\frac{x(1+x)}{(1-x)^3}$, rearranged slightly.

Exercise 1.4

```
> series(sin(x),x=Pi/2,12);
```

$$1-\frac{1}{2}\left(x-\frac{\pi}{2}\right)^2+\frac{1}{24}\left(x-\frac{\pi}{2}\right)^4-\frac{1}{720}\left(x-\frac{\pi}{2}\right)^6+\frac{1}{40320}\left(x-\frac{\pi}{2}\right)^8 - \frac{1}{3628800}\left(x-\frac{\pi}{2}\right)^{10}+\mathrm{O}\left(\left(x-\frac{\pi}{2}\right)^{12}\right) \quad \textbf{(23)}$$

```
> series(cos(x),x=0,12);
```

$$1-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}+\mathrm{O}(x^{12}) \quad \textbf{(24)}$$

If we call the first series $f(x)$ and the second series $g(x)$, then the relationship is that $f(x)=g\left(x-\frac{\mathrm{Pi}}{2}\right)$.
This is reasonable, because $f(x)$ approximates $\sin(x)$ and $g(x)$ approximates $\cos(x)$, and $\sin(x)=\cos\left(x-\frac{\mathrm{Pi}}{2}\right)$.

Exercise 2.1

```
> evalf[20](cos(ln(Pi+20)));
```

$$-0.99999999924368013310 \quad \textbf{(25)}$$

No one seems to have a good explanation for why this is so close to -1. It is probably just a coincidence.

Exercise 2.2

```
> f := (x) -> x^2-29/16;
```

$$f := x \rightarrow x^2-\frac{29}{16} \quad \textbf{(26)}$$

```
> f(0);
```

$$-\frac{29}{16} \quad \textbf{(27)}$$

```
> f(f(f(-1/4)));
```

$$-\frac{1}{4} \quad \textbf{(28)}$$

We see that $-\frac{1}{4}$ is sent to itself by the function $f(f(f(x)))$. There is a big theory about points fixed under
iterated application of functions like $f(x)$, which you can learn in the level 3 Chaos course.

Exercise 2.3

```
> a := (1+x)^5-3*(1+x)^4+5*(1+x)^3-3*(1+x)^2+3*(1+x)+3;
```

$$a := (1+x)^5-3(1+x)^4+5(1+x)^3-3(1+x)^2+6+3x \quad \textbf{(29)}$$

```
> b := (7*x^2-6*x-x^8)/(x-1)^2;
```

$$b := \frac{7x^2 - 6x - x^8}{(x-1)^2} \quad \textbf{(30)}$$

```
> simplify(a);
```

$$6 + 5x + 4x^2 + 3x^3 + 2x^4 + x^5 \quad \textbf{(31)}$$

```
> simplify(b);
```

$$-x\left(6 + 5x + 4x^2 + 3x^3 + 2x^4 + x^5\right) \quad \textbf{(32)}$$

It is clear from this that $b = -x\,a$.

Exercise 2.4

```
> y := ((x^12-1)*(x^2-1)/((x^6-1)*(x^4-1)))^(10);
```

$$y := \frac{\left(x^{12}-1\right)^{10}\left(x^2-1\right)^{10}}{\left(x^6-1\right)^{10}\left(x^4-1\right)^{10}} \quad \textbf{(33)}$$

```
> simplify(y);
```

$$\left(x^4 - x^2 + 1\right)^{10} \quad \textbf{(34)}$$

```
> coeff(simplify(y),x^6);
```

$$-210 \quad \textbf{(35)}$$

Exercise 2.5

```
> restart;
> solve({
   x^2 + y^2 + z^2 = 9,
   (x-1)^2 + (y-1)^2 + (z-1)^2 = 2,
   4*x^2+y*z = 2*x*y+2*x*z
  },{x,y,z});
```

$$\{x=1, y=2, z=2\}, \left\{x=\frac{8}{7}, y=\frac{11}{7}, z=\frac{16}{7}\right\}, \left\{x=\frac{8}{7}, y=\frac{16}{7}, z=\frac{11}{7}\right\} \quad \textbf{(36)}$$

We see that the only integer solution is $(x, y, z) = (1, 2, 2)$.

Exercise 2.6

```
> _EnvAllSolutions := true;
```

$$\_EnvAllSolutions := true \quad \textbf{(37)}$$

```
> solve(sin(theta)^2=3*cos(theta)^2,{theta});
```

$$\left\{\theta = \frac{2}{3}\pi - \frac{4}{3}\pi\_B1\sim + 2\pi\_Z1\sim\right\}, \left\{\theta = \frac{1}{3}\pi - \frac{2}{3}\pi\_B1\sim + 2\pi\_Z1\sim\right\} \quad \textbf{(38)}$$

```
> plot(sin(Pi*x)^2-3*cos(Pi*x)^2,x=-2..2);
```



The solutions are $\theta = \left(n + \frac{1}{3}\right)$ Pi and $\theta = \left(n - \frac{1}{3}\right)$ Pi for integers n.

Exercise 2.7

```
> _EnvExplicit := true;
```

$$\_EnvExplicit := true \quad \textbf{(39)}$$

```
> sols := solve({x^2-y^2=1,2*x*y=1},{x,y});
```

$$sols := \left\{x = \frac{\left(-2+2\sqrt{2}\right)^{3/2}}{4} + \sqrt{-2+2\sqrt{2}}, y = \frac{\sqrt{-2+2\sqrt{2}}}{2}\right\}, \left\{x = -\frac{\left(-2+2\sqrt{2}\right)^{3/2}}{4} - \sqrt{-2+2\sqrt{2}}, y = -\frac{\sqrt{-2+2\sqrt{2}}}{2}\right\}, \left\{x = \frac{\left(-2-2\sqrt{2}\right)^{3/2}}{4} + \sqrt{-2-2\sqrt{2}}, y = \frac{\sqrt{-2-2\sqrt{2}}}{2}\right\}, \left\{x = -\frac{\left(-2-2\sqrt{2}\right)^{3/2}}{4} - \sqrt{-2-2\sqrt{2}}, y = -\frac{\sqrt{-2-2\sqrt{2}}}{2}\right\} \quad \textbf{(40)}$$

```
> evalf(subs(sols[1],[x,y]));
```

$$[1.098684113, 0.4550898604] \quad \textbf{(41)}$$

```
> evalf(subs(sols[2],[x,y]));
```

$$[-1.098684113, -0.4550898604] \quad \textbf{(42)}$$

```
> evalf(subs(sols[3],[x,y]));
```

$$[-0.455089861\,I, 1.098684114\,I] \quad \textbf{(43)}$$

```
> evalf(subs(sols[4],[x,y]));
```

$$[0.455089861\, I,\ -1.098684114\, I] \quad (44)$$

Only the first of these is a pair of positive real numbers. Thus, the solution we want is as follows:

```
> sols[1];
```

$$\left\{x = \frac{\left(-2+2\sqrt{2}\right)^{3/2}}{4} + \sqrt{-2+2\sqrt{2}},\ y = \frac{\sqrt{-2+2\sqrt{2}}}{2}\right\} \quad (45)$$

```
> simplify(%);
```

$$\left\{x = \frac{\sqrt{-2+2\sqrt{2}}}{2} + \frac{\sqrt{-2+2\sqrt{2}}\,\sqrt{2}}{2},\ y = \frac{\sqrt{-2+2\sqrt{2}}}{2}\right\} \quad (46)$$

Exercise 2.8

```
> fsolve(x=log(x+20),x=3);
```

$$3.141633303 \quad (47)$$

This is related to Exercise 2.1. If the answer there were exactly -1, then the answer here would be Pi.